鶴巣 博行

株式会社ミガロ

RAD事業部 技術支援課

APIを利用した出力待ち行列情報の取得方法

IBM i のプログラミングはそこそこできるが
Delphi/400 はまだ使いこなせていないという人のために
このレポートを通じ、出力待ち行列処理の情報を画面上に表示し
表示した情報に対して処理を実行する簡単な方法を解説する。

略歴
1968 年 3 月 29 日 生
1990 年摂南大学 経営情報学部卒
2002 年株式会社ミガロ入社
2002 年 4 月システム事業部配属
2007 年 4 月 RAD 事業部配属

現在の仕事内容
現場（営業グループとシステム事業部）が活動しやすいような支援と、ミガロおよびミガロ製品の積極的なアピール活動を志している。

はじめに

Delphi/400 をご使用のユーザー様でアプリケーションの GUI 化はできているが、出力待ち行列処理等の管理作業のため、なかなか 5250 エミュレータを捨てきれないという声をよく聞く。

例えば､出力待ち行列の処理であれば、Delphi/400 には、SPOOLLIST400 や SPOOL400 という出力待ち行列処理を扱うための便利なコンポーネントが用意されている。それらのコンポーネントを駆使することで、出力待ち行列処理の制御を行うことができる。

Delphi/400 では、IBM i 上のデータを GUI 画面に表示させることが簡単にできる。また、GUI 画面上のデータをパラメータとし、IBM i のプログラムを呼び出すことも、そんなに難しいことではない。ゆえに、出力待ち行列処理の情報をデータベース化できさえすれば、簡単にその情報を GUI 画面上に表示することが可能だ。

また、出力待ち行列処理画面のオプションで行われる処理（例えば、3= 保留 4= 削除 5= 表示 6= 解放）は、結局のところ、選択されたファイルに対してコマンドを発行（保留：HLDSPLF 削 除：DLTSPLF 表 示：DSPSPLF 解放：RLSSPLF）しているだけである。なので、コマンド発行に必要な情報を付加し、例えば CALL400 コンポーネントを使用してプログラムで処理するか、もしくは AS400 コンポーネントで直接コマンドを発行してやれば、処理を行うことができる。これは、出力待ち行列処理画面のオプションで行われる処理と同じ処理である。

データベース化する方法

まずは、出力待ち行列処理の情報をデータベース化する方法である。

コマンドで実現できないかといろいろ調べたのだが、直接データベース化できそうなコマンドは見つけられなかった。

そこで、コマンドとプログラムとの組み合わせで、何とかデータベース化する方法を思いついた。だか、やり方が強引であることと取得できる情報量が少ないことから、あまり実用性がないと判断した。【図 1】

IBM アンサーラインや Web などを利用し、他に実現方法がないか調査したところ、API を使用し、出力待ち行列処理が取得できることがわかった。

APIとは

API（アプリケーション・プログラミング・インタフェース、Application Programming Interface）について、IT 用語辞典（e-Words）の説明をそのまま引用する。

「あるプラットフォーム（OS やミドルウェア）向けのソフトウェアを開発する際に使用できる命令や関数の集合のこと。また、それらを利用するためのプログラム上の手続きを定めた規約の集合。

個々のソフトウェアの開発者がソフトウェアの持つすべての機能をプログラミングするのは困難で無駄が多いため、多

図1　コマンドとプログラムのを組合わせた出力待ち行列のデータベース化

手順１．WRKOUTQ コマンドでデータベース化したい OUTQ の一覧を印刷する。

WRKOUTQ OUTQ(QGPL/QPRINT) OUTPUT(*PRINT)



くのソフトウェアが共通して利用する機能は、OSやミドルウェアなどの形でまとめて提供されている。

個々の開発者は規約に従ってその機能を「呼び出す」だけで、自分でプログラミングすることなくその機能を利用したソフトウェアを作成することができる」

OSのコマンドでサポートされない機能をサポートする、ということが主要な目的のようだ。IBM iのOSで用意されているAPIは、主にオブジェクトの属性取り出しやオブジェクトの一覧処理だが、その他にも数字の編集などがある。(APIを使いこなしているわけではないので、もっと他にもたくさんあるかもしれない)。

難点は、わかりやすいマニュアルがない点だが、使用方法はパラメータで要求情報をAPIプログラムに渡し、結果がパラメータで返ってくるといったものなので、各々のパラメータの意味がわかれば使用できる。

コーディング方法

まず、APIを使用して取得できる情報のDDSを作成する。データベース名はUSRSPLPとし、用意するフィールドは［図2］のとおりである。【図2】

あとは、このデータベースにデータを落とすプログラムを作成する。ソースコードは［図3］のようになる。これを参考にしながら説明を読んでほしい。【図3】

①ユーザー・スペースの作成

ユーザー・スペースとは、APIで使用する新しいオブジェクト（オブジェクトタイプは＊USRSPC）で、一覧形式で予め何レコードの結果が出るかわからないものを検索する場合に使用される。

といっても、難しい作業をする必要はなく、ユーザー・スペース作成用のAPI（QUSCRTUS）が用意されている。なので、そのAPIを規定のパラメータに値をセットし、呼び出せば作成してくれる。

今回は、ライブラリー：QTEMPに、ユーザー・スペース：SPLINFを、ユーザー・スペース初期サイズ：1024で作成するようにしている。

ユーザー・スペースは「総称見出し」と「リストセクション」から構成される。(自分なりにわかりやすく表現すると、ヘッダーと明細の関係みたいな感じではないかと思う)。

②ユーザー・スペースに出力待ち行列のデータを展開

次に、①で用意したユーザー・スペースに、出力待ち行列の情報を展開する。

使用するAPIはQUSLSPLである。QUSLSPLに、以下のパラメータをセットし実行する。

・ユーザー・スペース（20桁）
・フォーマット形式（8桁）
・出力待ち行列のユーザー（10桁）
・OUTQ（20桁）
・フォームタイプ（10桁）
・ユーザーデータ（10桁）

今回はすべてを対象にしているが、このパラメータのセットの仕方により、取得したい出力待ち行列を制御することが可能である。

例えば、待ち行列QPRINT分だけを取得したい場合は、パラメータのOUTQに“QRINT △△△△△ QGPL △△△△△△”（三角はスペース）と指定する。

データの展開が失敗した場合は（ソース上では標識95がONになる）、処理が中断される。

③総称見出しからデータを取り出す

次に、②で取得した情報から総称見出しの情報を取得する。(取得したい項目は、リスト・データ・セクションのオフセットと、リスト・データ・セクションのサイズと、リストセクションに書き出された数)。

使用するAPIはQUSRTVUSである。QUSRTVUSに、以下のパラメータをセットし実行する。

・ユーザー・スペース（20桁）
・開始位置（4桁）
・取得数（4桁）
・結果を入れるフィールド

総称見出しのレイアウトは［図4］のとおりである。レイアウトを見るとオフセット10進数が0から始まっている。ただし、QUSRTVUS APIの開始位置を決定するためには、オフセット値に1を加算しなければならないので注意しよう。

例えば、レイアウト中の情報の状況を使用したい場合、開始位置は104（オフセット10進数:103 + 1）となる。【図4】

今回の場合、開始位置を125（リスト・データ・セクションのオフセット）から16バイト分（各項目のサイズまで）の情報を取得する。取得した情報は、I仕様書のDSで分解する。

総称見出しからのデータ取り出しが失敗した場合は（ソース上では標識95がONになる）、処理が中断する。また、取得したリスト項目の数(フィールド名:NOENTH）が0の場合は、データがないということなので、この場合も処理を中断する。

④リストセクション読み込みの初期化

⑤で使用するAPIのパラメータ値の初期設定を行う。

③で取得したリスト･データ･セクションのオフセットに、1を足し、パラメータ値にセットする。また、同じく③で取得したリスト・データ・セクションのサイズも、パラメータ値にセットする。

⑤リストセクションの各項目の取り出し

②で取得した出力待ち行列の情報を、リストセクションから取得する。

使用するAPIは③で使用したのと同じQUSRTVUSである。パラメータSF0100に結果がセットされる。取得できる情報は、以下のものである。

・ユーザー名（USRNML）
・OUTQとライブラリー（OUTQL）
・フォームタイプ名（FRMTYL）
・ユーザーデータ（USRDTL）
・内部JOB確認者（JOBID）
・内部スプール・ファイル確認者（PLFID）

次に、QUSRTVUSで取得した内部JOB確認者と内部スプール・ファイル確認者をもとに、追加の属性を取得する。(サブルーチン @ATR)

使用するAPIはQUSRSPLAである。QUSRSPLAに、以下のパラメータを

手順２．印刷したスプールファイルを CPYSPLF でデータベース化する。

ｉ．データベースを作成する。
CRTPF FILE(QTEMP/PRTDB) RCDLEN(132) IGCDTA(*YES)
ⅱ．CPYSPLF コマンド実行
CPYSPLF FILE(QPRTSPLQ) TOFILE(QTEMP/PRTDB)



手順３．プログラムでデータ化した一覧を項目毎に区切り DDS をきったデータベースに移行する。

ｉ．１レコードずつ手順２で作成したファイルを読み込む。
ⅱ．３桁目が英字（A ～ Z）のレコードを対象とする。
ⅲ．３～１２桁目をファイル名、１４～２３桁目までをユーザー名といった手順で各フィールド毎に切り分け、フィールドにセットする。

セットし実行する。

・結果を受け取るフィールド
・結果を受け取るフィールドのサイズ（4桁）
・形式名（8桁）
・修飾ジョブ名（26桁）
・内部ジョブ識別コード（16桁）
・内部スプール･ファイル識別コード（16桁）
・スプール・ファイル名（10桁）
・スプール・ファイル番号（4桁）

取得できる情報は、USRSPLPのレイアウト［図2］の情報になる。

⑥項目の書き出し

⑤で取得した情報を、データベースUSRSPLPの各フィールドにセットし、レコードを書き出す。

コーディング方法を①～⑥に述べた。⑤と⑥の作業を、③で取得したリスト項目の数（フィールド名：NOENTH）分繰り返すことで、出力待ち行列の情報をデータベース化できる。

あとは、データベース化した情報を、Delphi/400を使用して、GUI画面上にグリッド等を利用し表示する。出力待ち行列を照会するだけであれば、これで完成だ。

出力待ち行列への処理実行

出力待ち行列に対して処理を実行させたい場合は、処理ごとのボタン（保留、削除、表示、解放）を用意する。

各ボタンから呼び出されるプログラムは、共通でも、ボタンごとにCall400コンポーネントでプログラムを呼び出しても、AS400コンポーネントでリモートコマンドを発行してもかまわない。

今回は、共通のCLPを使用する例をあげる。【図5】

CLP実行時パラメータを以下のようにする。

・処理区分（1桁）
　保留：3　削除：4　表示：5　送信：6
・スプール・ファイル名（10桁）
・JOB名（10桁）
・ユーザー名（10桁）
・ジョブ番号（6桁）
・SPLF番号（6桁）
・JOBシステム名（8桁）

各ボタン押下時、押されたボタンの処理区分（保留：3　削除：4　表示：5　送信：6）をセットし、グリッドで選択されているスプール・ファイルからパラメータに必要な情報をセットし、CLPを呼び出すロジックを組み込めば、プログラムが完成する。IBM iの出力待ち行列の処理と、同等の仕組みを持ったプログラムである。

まとめ

ここまで述べてきたように、出力待ち行列の処理であれば、

・出力待ち行列表示用画面（Delphi/400）
・出力待ち行列データ作成プログラム（RPG）
・出力待ち行列処理用のプログラム（CLP）

を組み合わせれば、簡単に作成することができる。

APIを使用する部分は若干難しいかもしれないが、APIの使用ルールは形式が定まっているので、そのとおり作成すれば問題ない。

IBM iには、今回取り上げた出力待ち行列処理以外にもいろいろな管理画面がある。結局のところ、どこからかデータを取得し、画面に表示し、何らかの処理を実行する場合、そのデータ値をパラメータとしコマンドを発行しているだけである。なので、データの取得方法さえわかれば、同様の処理をGUI化することはそう難しくないと思う。

今後も便利な情報を提供するために、Delphi/400をより有効に使用できる手段を見つけたいと思っている。

M

現在の仕事内容（詳細）
セミナーやフェアの開催と集客、お客様への製品説明や提案といった営業推進活動、および、製品とお客様からの問い合わせへの対応などの顧客サポート活動を行っている。また、製品出荷やメンテナンス更新管理、売上管理などの営業事務管理と、ホームページ更新やメールマガジンといった宣伝活動も行っている。

図2　APIを使用して取得できる出力待ち行列情報

```
A*----------------------------------------------------------------*
A*  FILE ID.           :  USRSPLP                                  *
A*  DESCRIPTION        :  スプールファイル属性                     *
A*  FORMAT ID.         :  USRSPLR                                  *
A*                                                                 *
A*----------------------------------------------------------------*
A          R USRSPLR
A            SPL001        4B 0       COLHDG('返り値バイト数')
A            SPL002        4B 0       COLHDG('利用可能バイト数')
A            SPL003        16A        COLHDG('内部 JOB 確認者')
A            SPL004        16A        COLHDG('内部 SPLF 確認者')
A            SPL005        10A        COLHDG('ジョブ名')
A            SPL006        10A        COLHDG('ユーザー名')
A            SPL007        6A         COLHDG('ジョブ番号')
A            SPL008        10A        COLHDG('SPLF 名')
A            SPL009        4B 0       COLHDG('SPLF 番号')
A            SPL010        10A        COLHDG('形式タイプル名')
A            SPL011        10A        COLHDG('ユーザー指定データ')
A            SPL012        10A        COLHDG('状態ファイル名')
A            SPL013        10A        COLHDG('利用可能ファイル')
A            SPL014        10A        COLHDG('書込前 HOLD FILE')
A            SPL015        10A        COLHDG('記述後ファイル保存')
A            SPL016        4B 0       COLHDG('合計ページイル保存')
A            SPL017        4B 0       COLHDG('書き込み中ページ')
A            SPL018        4B 0       COLHDG('開始ページ')
A            SPL019        4B 0       COLHDG('終了ページ')
A            SPL020        4B 0       COLHDG('最終印刷ページ')
A            SPL021        4B 0       COLHDG('再印刷')
A            SPL022        4B 0       COLHDG('トータルコピー数')
A            SPL023        4B 0       COLHDG('生成用コピー残')
A            SPL024        4B 0       COLHDG('行／インチ')
A            SPL025        4B 0       COLHDG('文字／インチ')
A            SPL026        2A         COLHDG('優先出力')
A            SPL027        10A        COLHDG('出力キュー名名')
A            SPL028        10A        COLHDG('出力 Q ライブラリ名')
A            SPL029        7A         COLHDG('日付作成ファイル')
A            SPL030        6A         COLHDG('時間作成ファイル')
A            SPL031        10A        COLHDG('装置ファイル名')
A            SPL032        10A        COLHDG('装置ファイル LIB')
A            SPL033        10A        COLHDG('FILE を開いた PGM')
A            SPL034        10A        COLHDG('')
A            SPL035        15A        COLHDG('経理基準')
A            SPL036        30A        COLHDG('テキスト印刷')
A            SPL037        4B 0       COLHDG('レコード長')
A            SPL038        4B 0       COLHDG('最大レコード数')
A            SPL039        10A        COLHDG('装置タイプ')
A            SPL040        10A        COLHDG('プリンタ装置タイプ')
A            SPL041        12A        COLHDG('ドキュメント名')
A            SPL042        64A        COLHDG('フォルダ名ル名')
A            SPL043        8A         COLHDG('SYSTEM36 手続き名')
A            SPL044        10A        COLHDG('印刷精度')
A            SPL045        1A         COLHDG('印刷不可能文字変換')
A            SPL046        1A         COLHDG('文字置換')
A            SPL047        4B 0       COLHDG('ページ長')
A            SPL048        4B 0       COLHDG('ページ幅')
A            SPL049        4B 0       COLHDG('分割数')
A            SPL050        4B 0       COLHDG('オーバフロー行番号')
A            SPL051        10A        COLHDG('DBCS データル名')
A            SPL052        10A        COLHDG('DBCS 拡張文字')
A            SPL053        10A        COLHDG('DBCS 変更間隔')
A            SPL054        10A        COLHDG('DBCS 文字回転')
A            SPL055        4B 0       COLHDG('DBCS 文字／インチ')
A            SPL056        10A        COLHDG('GRAPHIC 文字 SET')
A            SPL057        10A        COLHDG('コードページ')
A            SPL058        10A        COLHDG('定義名形式')
A            SPL059        10A        COLHDG('ライブラリ定義形式')
A            SPL060        4B 0       COLHDG('ソース引出ル名')
A            SPL061        10A        COLHDG('プリンタフォント')
A            SPL062        6A         COLHDG('SYS36 SPLF 確認者')
A            SPL063        4B 0       COLHDG('ページ回転')
A            SPL064        4B 0       COLHDG('正当化')
A            SPL065        10A        COLHDG('両面印刷')
A            SPL066        10A        COLHDG('重複レコード')
A            SPL067        10A        COLHDG('制御文字')
A            SPL068        10A        COLHDG('整列形式')
A            SPL069        10A        COLHDG('印刷品質')
A            SPL070        10A        COLHDG('形式材料')
A            SPL071        71A        COLHDG('ボリューム（配列）')
A            SPL072        17A        COLHDG('FILE ラベル確認者')
A            SPL073        10A        COLHDG('交換タイプ')
A            SPL074        10A        COLHDG('文字コード')
A            SPL075        4B 0       COLHDG('合計レコード数')
A            SPL076        4B 0       COLHDG('倍角文字')
A            SPL077        10A        COLHDG('前面オーバーレイ')
A            SPL078        10A        COLHDG('ライブラリー')
A            SPL079        15P 5      COLHDG('下方向オフセット')
A            SPL080        15P 5      COLHDG('横方向オフセット')
A            SPL081        10A        COLHDG('背面オーバーレイ')
A            SPL082        10A        COLHDG('ライブラリー')
A            SPL083        15P 5      COLHDG('下方向オフセット')
A            SPL084        15P 5      COLHDG('横方向オフセット')
A            SPL085        10A        COLHDG('ユニット寸法')
A            SPL086        10A        COLHDG('ページ定義名')
A            SPL087        10A        COLHDG('ページ定義 LIB 名')
A            SPL088        10A        COLHDG('ライン間隔')
A            SPL089        15P 5      COLHDG('ポイントサイズ')
A            SPL090        15P 5      COLHDG('下へのマージン')
A            SPL091        15P 5      COLHDG('横へのマージン')
A            SPL092        15P 5      COLHDG('下へのマージン')
A            SPL093        15P 5      COLHDG('横へのマージン')
A            SPL094        15P 5      COLHDG('ページ長')
A            SPL095        15P 5      COLHDG('ページ幅')
A            SPL096        10A        COLHDG('測定方法イル名')
A            SPL097        1A         COLHDG('高度関数印刷資源')
A            SPL098        10A        COLHDG('文字セット名')
A            SPL099        10A        COLHDG('文字セット LIB 名')
A            SPL100        10A        COLHDG('コードページ名')
A            SPL101        10A        COLHDG('コードページ LIB')
A            SPL102        10A        COLHDG('コードフォント名')
A            SPL103        10A        COLHDG('コードフォント LIB')
A            SPL104        10A        COLHDG('DBCS コードフォント')
A            SPL105        10A        COLHDG('DBCS コード FONT LIB')
A            SPL106        10A        COLHDG('ユーザー定義 FILE')
A            SPL107        10A        COLHDG('縮小出力')
A            SPL108        1A         COLHDG('固定バック')
A            SPL109        4B 0       COLHDG('BIN 出力 ル名')
A            SPL110        4B 0       COLHDG('CCSID ァイル名')
A            SPL111        100A       COLHDG('ユーザ定義テキスト')
A            SPL112        8A         COLHDG('ファイル作成 SYSTEM')
A            SPL113        8A         COLHDG('ファイル作成 ID')
A            SPL114        10A        COLHDG('ファイル作成者')
A            SPL115        2A         COLHDG('保存')
A            SPL116        4B 0       COLHDG('ユーザ定義 OPT')
A            SPL117        4B 0       COLHDG('戻値ユーザ定義 OPT')
A            SPL118        4B 0       COLHDG('ユーザ定義 ENTRY')
A            SPL119        255A       COLHDG('ユーザ定義データ')
A            SPL120        10A        COLHDG('ユーザ定義 OBJ')
A            SPL121        10A        COLHDG('ユーザ定義 OBJ LIB')
A            SPL122        10A        COLHDG('ユーザー OBJ TYPE')
A            SPL123        3A         COLHDG('保存')
A            SPL124        15P 5      COLHDG('文字セットサイ')
A            SPL125        15P 5      COLHDG('コードフォント SIZE')
A            SPL126        15P 5      COLHDG('DBCSCODE FONTSIZE')
A            SPL127        4B 0       COLHDG('補助記憶装置プール')
A            SPL128        4B 0       COLHDG('SPLF サイズ')
A            SPL129        4B 0       COLHDG('SPLF サイズ乗数')
A            SPL130        4B 0       COLHDG('INTERNET 印刷')
A            SPL131        1A         COLHDG('SPLF生成セキュリティ')
A            SPL132        1A         COLHDG('SPLF 生成認証方法')
A            SPL133        7A         COLHDG('SPLF 処理開始者日付')
A            SPL134        6A         COLHDG('SPLF 処理開始者時間')
A            SPL135        7A         COLHDG('SPLF 処理完成者日付')
A            SPL136        6A         COLHDG('SPLF 処理完成者時間')
A            SPL137        8A         COLHDG('ジョブシステム名')
```

図3　APIを使用して出力待ち行列情報を取得するプログラム

```
H           Y
F*************************************************************
F*DEFINE
F*************************************************************
FUSRSPLP O  E           K        DISK
F*
IUSRSPC     DS
I                                        1  10 USNAME
I                                       11  20 USLIB
IGENHED     DS
I                                    B   1   4ODTLOFS
I                                    B   5   8OLSTSIZ
I                                    B   9  12ONOENTH
I                                    B  13  16ODTLSIZ
I           DS
I                                    B   1   4OUSSIZE
I                                    B   5   8OLENDTA
I                                    B   9  12OSTRPOS
I           DS
I                                    B   1   4ORCVLEN
I                                    B   5   8OSPLF#
ISF0100     DS
I                                        1  10 USRNML
I                                       11  30 OUTQL
I                                       11  20 OUTQNL
I                                       21  30 OUTQLL
I                                       31  40 FRMTYL
I                                       41  50 USRDTL
I                                       51  66 JOBID
I                                       67  82 SPLFID
ISA0100     DS
I                                    B   1   4OBYTESR
I                                    B   5   8OBYTESA
I                                        9  24 LJOBID
I                                       25  40 LSPLID
I                                       41  50 JOBNAM
I                                       51  60 USRNAM
I                                       61  66 JOBNUM
I                                       67  76 FILNAM
I                                    B  77  80OFILNUM
I                                       81  90 FRMTYP
I                                       91 100 USRDTA
I                                      101 110 STAT
I                                      111 120 FILAVA
I                                      121 130 HOLDF
I                                      131 140 SAVF
I                                    B 141 144OTOTALP
I                                    B 145 148OBEGWRT
I                                    B 149 152OSTART
I                                    B 153 156OEND
I                                    B 157 160OLASTPG
I                                    B 161 164ORESTAT
I                                    B 165 168OCOPYS
I                                    B 169 172OPRODUC
I                                    B 173 176OLINPER
I                                    B 177 180OCHRPER
I                                      181 182 PRIORT
I                                      183 192 OUTQNM
I                                      193 202 OUTQLB
I                                      204 209 DATFOP
I                                      210 215 TIME
I                                      216 225 DVFINM
I                                      226 235 DVFLNM
I                                      236 245 POFINM
I                                      246 255 POFLNM
I                                      256 270 ACCCD
I                                      271 300 TEXT
I                                    B 301 304ORECLEN
I                                    B 305 308OMAXREC
I                                      309 318 DEVTYP
I                                      319 328 PRDVTP
I                                      329 340 DOCNAM
I                                      341 404 FLDNAM
I                                      405 412 S36PNM
I                                      413 422 PRFIDE
I                                      423 423 REUNCH
I                                      424 424 REMCHR
I                                    B 425 428OPAGLEN
I                                    B 429 432OPAGWID
I                                    B 433 436ONMSEP
I                                    B 437 44000VFLN
I                                      441 450 DBCSDA
I                                      451 460 DBCSEX
I                                      461 470 DBCSSO
I                                      471 480 DBCSCR
I                                    B 481 484ODBCSCI
I                                      485 494 GRCHS
I                                      495 504 CODPAG
I                                      505 514 FOMDEF
I                                      515 524 FOMDEL
I                                    B 525 528OSOUDRA
I                                      529 538 PRFONT
I                                      539 544 S36SPL
I                                    B 545 548OPAGROT
I                                    B 549 552OJUST
I                                      553 562 PRTOBT
I                                      563 572 FOLREC
I                                      573 582 CNTCHR
I                                      583 592 ALIFOR
I                                      593 602 PRTQUA
I                                      603 612 FOMFED
I                                      613 683 VOL
I                                      684 700 FILLAB
I                                      701 710 EXCHTP
I                                      711 720 CHRCOD
I                                    B 721 724OTOLREC
I                                    B 725 728OMULTUP
I                                      729 738 FROVRN
I                                      739 748 FROVRL
I                                    P 749 756SFROVRO
I                                    P 757 764SFROVRA
I                                      765 774 BAOVRN
I                                      775 784 BAOVRL
I                                    P 785 792SBAOVRO
I                                    P 793 800SBAOVRA
I                                      801 810 UNTOME
I                                      811 820 PAGDEF
I                                      821 830 PAGDEL
I                                      831 840 LINSPC
I                                    P 841 848SPITSIZ
I                                    P 849 856SFRTMAG
I                                    P 857 864SFRTMAA
I                                    P 865 872SBACMAG
I                                    P 873 880SBACMAA
I                                    P 881 888SLEGOPG
I                                    P 889 896SWIDTH
I                                      897 906 MESMET
I                                      907 907 ADVFUN
I                                      908 917 CHRSET
I                                      918 927 CHRSEL
I                                      928 937 CODPAN
I                                      938 947 CODPAL
I                                      948 957 CODFNT
I                                      958 967 CODFNN
I                                      968 977 DBCSNM
I                                      978 987 DBCSLI
I                                      988 997 USRDEF
I                                     9981007 REDOUT
I                                    10081008 CONST
I                                    B100910120BIN
I                                    B101310160CCSID
I                                     10171116 USRTXT
I                                     11171124 SYSCRT
I                                     11251132 IDCRT
I                                     11331142 USRFIL
I                                     11431144 RESERV
```

```
I                                 B1145114800FFSET
I                                 B1149115200PT
I                                 B1153115600LEN0PT
I                                  11571411 DEFDAT
I                                  14121421 OBJNAM
I                                  14221431 OBJLNM
I                                  14321441 OBJTYP
I                                  14421444 RES
I                                 P1445145250NTSI
I                                 P1453146050FNTPOS
I                                 P1461146850BCSSZ
I                                 B1469147200AUXST0
I                                 B1473147600SPLFSI
I                                 B1477148000SPLFSM
I                                 B1481148400INTPRT
I                                  14851485 SECMET
I                                  14861486 AUTMET
I                                  14871493 DATWTR
I                                  14941499 TIMWTR
I                                  15001506 DCOMP
I                                  15071512 TCOMP
I                                  15131520 JOBSYS
I***
I             'MATSUP    QUSRSYS   'C         OUTQ#
C*****************************************************
C*            M A I N - R O U T I N E               :
C*****************************************************
C*                                                  :
C* ①ユーザー・スペースの作成                        :
C*                                                  :
C                     Z-ADD1024     USSIZE          :
C                     MOVEL'SPLINF' USNAME          :
C                     MOVEL'QTEMP'  USLIB           :
C*                                                  :
C                     CALL 'QUSCRTUS'           95  :
C                     PARM          USRSPC          :
C                     PARM 'SPL'    EXTATR 10       :
C                     PARM          USSIZE          :
C                     PARM X'00'    USINIT  1       :
C                     PARM '*ALL'   USAUTH 10       :
C                     PARM          USTEXT 50       :
C*                                                  :
C* ②ユーザー・スペースにデータを展開                :
C*                                                  :
C                     CALL 'QUSLSPL'            95  :
C                     PARM          USRSPC          :
C                     PARM 'SPLF0100'FMTRCD  8      :
C                     PARM 'USER    'USER   10      :
C                     PARM '*ALL'   OUTQ   20       :
C                     PARM '*ALL'   FRMTYP 10       :
C                     PARM '*ALL'   USRDTA 10       :
C*                                                  :
C           *IN95     CABEQ*ON      $END            :
C*                                                  :
C* ③総称見出しからデータを取り出す                  :
C*                                                  :
C                     CALL 'QUSRTVUS'           95  :
C                     PARM          USRSPC          :
C                     PARM 125      STRPOS          :
C                     PARM 16       LENDTA          :
C                     PARM          GENHED          :
C*                                                  :
C           *IN95     CABEQ*ON      $END            :
C           NOENTH    CABEQ0        $END            :
C*                                                  :
C* ④リストセクション読み込みの初期化                :
C*                                                  :
C                     Z-ADD1        W1CNT   50      :
C*                                                  :
C                     Z-ADDDTL0FS   STRPOS          :
C                     ADD  1        STRPOS          :
C                     Z-ADDDTLSIZ   LENDTA          :
C*                                                  :
C* ⑤リストセクションの各項目取り出し                :
C*                                                  :
B001 C           W1CNT     DOWLENOENTH              :
 001 C*                                             :
 001 C                     CALL 'QUSRTVUS'           95  :
 001 C                     PARM          USRSPC          :
 001 C                     PARM          STRPOS          :
 001 C                     PARM          LENDTA          :
 001 C                     PARM          SF0100          :
 001 C*                                             :
 001 C                     EXSR @ATR                :
     C* ⑥項目の書き出し                                :
     C*           OUTQNM    IFEQ 'FAXQ'
     C                     EXSR @WRITE
     C*                    ENDIF
 001 C*                                             :
 001 C                     ADD  1        W1CNT           :
 001 C                     ADD  DTLSIZ   STRPOS          :
 001 C*                                             :
E001 C                     ENDDO                    :
     C*                                             :
     C           $END      TAG                      :
     C*                                             :
     C                     MOVE *ON      *INLR           :
     C                     RETRN                    :
     C*****************************************************
     C*            S U B  - R O U T I N E
     C*****************************************************
     C*---------------------------------------------------*
     C           @ATR      BEGSR
     C*---------------------------------------------------*
     C                     Z-ADD1520     RCVLEN          :
     C                     MOVE *BLANK   JOBINF 26       :
     C                     MOVE *BLANK   SPLFNM 10       :
     C                     Z-ADD0        SPLF#           :
     C*                                             :
     C                     CALL 'QUSRSPLA'           95  :
     C                     PARM          SA0100          :
     C                     PARM          RCVLEN          :
     C                     PARM 'SPLA0100'FMTATR  8      :
     C                     PARM '*INT'   JOBINF          :
     C                     PARM JOBID    LJOBID          ::
     C                     PARM SPLFID   LSPLID          :
     C                     PARM '*INT'   SPLFNM          :
     C                     PARM          SPLF#           :
     C*
     C* ここで、スプールの各属性が取り込まれる。
     C*                                             :
     C                     ENDSR                    :
     C*---------------------------------------------------*
     C           @WRITE    BEGSR
     C*---------------------------------------------------*
     C                     MOVELBYTESR   SPL001
     C                     MOVELBYTESA   SPL002
     C                     MOVELLJOBID   SPL003
     C                     MOVELLSPLID   SPL004
     C                     MOVELJOBNAM   SPL005
     C                     MOVELUSRNAM   SPL006
     C                     MOVELJOBNUM   SPL007
     C                     MOVELFILNAM   SPL008
     C                     MOVELFILNUM   SPL009
     C                     MOVELFRMTYP   SPL010
     C                     MOVELUSRDTA   SPL011
     C                     MOVELSTAT     SPL012
     C                     MOVELFILAVA   SPL013
     C                     MOVELHOLDF    SPL014
     C                     MOVELSAVF     SPL015
     C                     MOVELTOTALP   SPL016
     C                     MOVELBEGWRT   SPL017
     C                     MOVELSTART    SPL018
     C                     MOVELEND      SPL019
     C                     MOVELLASTPG   SPL020
     C                     MOVELRESTAT   SPL021
     C                     MOVELCOPYS    SPL022
     C                     MOVELPRODUC   SPL023
     C                     MOVELLINPER   SPL024
     C                     MOVELCHRPER   SPL025
```

```
C                   MOVELPRIORT     SPL026
C                   MOVELOUTQNM     SPL027
C                   MOVELOUTQLB     SPL028
C                   MOVELDATFOP     SPL029
C                   MOVELTIME       SPL030
C                   MOVELDVFINM     SPL031
C                   MOVELDVFLNM     SPL032
C                   MOVELPOFINM     SPL033
C                   MOVELPOFLNM     SPL034
C                   MOVELACCCD      SPL035
C                   MOVELTEXT       SPL036
C                   MOVELRECLEN     SPL037
C                   MOVELMAXREC     SPL038
C                   MOVELDEVTYP     SPL039
C                   MOVELPRDVTP     SPL040
C                   MOVELDOCNAM     SPL041
C                   MOVELFLDNAM     SPL042
C                   MOVELS36PNM     SPL043
C                   MOVELPRFIDE     SPL044
C                   MOVELREUNCH     SPL045
C                   MOVELREMCHR     SPL046
C                   MOVELPAGLEN     SPL047
C                   MOVELPAGWID     SPL048
C                   MOVELNMSEP      SPL049
C                   MOVELOVFLN      SPL050
C                   MOVELDBCSDA     SPL051
C                   MOVELDBCSEX     SPL052
C                   MOVELDBCSSO     SPL053
C                   MOVELDBCSCR     SPL054
C                   MOVELDBCSCI     SPL055
C                   MOVELGRCHS      SPL056
C                   MOVELCODPAG     SPL057
C                   MOVELFOMDEF     SPL058
C                   MOVELFOMDEL     SPL059
C                   MOVELSOUDRA     SPL060
C                   MOVELPRFONT     SPL061
C                   MOVELS36SPL     SPL062
C                   MOVELPAGROT     SPL063
C                   MOVELJUST       SPL064
C                   MOVELPRTOBT     SPL065
C                   MOVELFOLREC     SPL066
C                   MOVELCNTCHR     SPL067
C                   MOVELALIFOR     SPL068
C                   MOVELPRTQUA     SPL069
C                   MOVELFOMFED     SPL070
C                   MOVELVOL        SPL071
C                   MOVELFILLAB     SPL072
C                   MOVELEXCHTP     SPL073
C                   MOVELCHRCOD     SPL074
C                   MOVELTOLREC     SPL075
C                   MOVELMULTUP     SPL076
C                   MOVELFROVRN     SPL077
C                   MOVELFROVRL     SPL078
C                   Z-ADDFROVRO     SPL079
C                   Z-ADDFROVRA     SPL080
C                   MOVELBAOVRN     SPL081
C                   MOVELBAOVRL     SPL082
C                   Z-ADDBAOVRO     SPL083
C                   Z-ADDBAOVRA     SPL084
C                   MOVELUNTOME     SPL085
C                   MOVELPAGDEF     SPL086
C                   MOVELPAGDEL     SPL087
C                   MOVELLINSPC     SPL088
C                   Z-ADDPITSIZ     SPL089
C                   Z-ADDFRTMAG     SPL090
C                   Z-ADDFRTMAA     SPL091
C                   Z-ADDBACMAG     SPL092
C                   Z-ADDBACMAA     SPL093
C                   Z-ADDLEGOPG     SPL094
C                   Z-ADDWIDTH      SPL095
C                   MOVELMESMET     SPL096
C                   MOVELADVFUN     SPL097
C                   MOVELCHRSET     SPL098
C                   MOVELCHRSEL     SPL099
C                   MOVELCODPAN     SPL100
C                   MOVELCODPAL     SPL101
C                   MOVELCODFNT     SPL102
C                   MOVELCODFNN     SPL103
C                   MOVELDBCSNM     SPL104
C                   MOVELDBCSLI     SPL105
C                   MOVELUSRDEF     SPL106
C                   MOVELREDOUT     SPL107
C                   MOVELCONST      SPL108
C                   MOVELBIN        SPL109
C                   MOVELCCSID      SPL110
C                   MOVELUSRTXT     SPL111
C                   MOVELSYSCRT     SPL112
C                   MOVELIDCRT      SPL113
C                   MOVELUSRFIL     SPL114
C                   MOVELRESERV     SPL115
C                   MOVELOFFSET     SPL116
C                   MOVELOPT        SPL117
C                   MOVELLENOPT     SPL118
C                   MOVELDEFDAT     SPL119
C                   MOVELOBJNAM     SPL120
C                   MOVELOBJLNM     SPL121
C                   MOVELOBJTYP     SPL122
C                   MOVELRES        SPL123
C                   Z-ADDPONTSI     SPL124
C                   Z-ADDFNTPOS     SPL125
C                   Z-ADDDBCSSZ     SPL126
C                   MOVELAUXSTO     SPL127
C                   MOVELSPLFSI     SPL128
C                   MOVELSPLFSM     SPL129
C                   MOVELINTPRT     SPL130
C                   MOVELSECMET     SPL131
C                   MOVELAUTMET     SPL132
C                   MOVELDATWTR     SPL133
C                   MOVELTIMWTR     SPL134
C                   MOVELDCOMP      SPL135
C                   MOVELTCOMP      SPL136
C                   MOVELJOBSYS     SPL137
C                   WRITEUSRSPLR
C*
C                   ENDSR                   :
```

図4　総称見出しのレイアウト

```
  0 CHAR(64)  ユーザー域
 64 BINARY(4) 総称見出しのサイズ
 68 CHAR(4)   構造のリリースおよびレベル
 72 CHAR(8)   形式名
 80 CHAR(10)  使用された API
 90 CHAR(13)  作成された日時
103 CHAR(1)   情報の状況
104 BINARY(4) 使用されるユーザー空間のサイズ
108 BINARY(4) 入力パラメーター・セクションへのオフセット
112 BINARY(4) 入力パラメーター・セクションのサイズ
116 BINARY(4) 見出しセクションのオフセット
120 BINARY(4) 見出しセクションのサイズ
124 BINARY(4) リスト・データ・セクションのオフセット
128 BINARY(4) リスト・データ・セクションのサイズ
132 BINARY(4) リスト項目の数
136 BINARY(4) 各項目のサイズ
140 BINARY(4) リスト項目のデータの CCSID
144 CHAR(2)   国別 ID
147 CHAR(3)   言語 ID
149 CHAR(1)   サブセットされたリスト標識
150 CHAR(42)  予約済み
```

図5　出力待ち行列処理実行プログラム

```
PGM         PARM(       +
                &PRMKBN +
                &PRMSPF +
                &PRMJOB +
                &PRMUSR +
                &PRMJNO +
                &PRMSFN )

              /*  変数定義 */
            DCL  VAR(&PRMKBN) TYPE(*CHAR) LEN(1)   /*  処理区分         */
              /* ３＝保留　４＝削除　５＝表示　６＝送信 */
            DCL  VAR(&PRMSPF) TYPE(*CHAR) LEN(10)  /*  ＳＰＬＦ名       */
            DCL  VAR(&PRMJOB) TYPE(*CHAR) LEN(10)  /*  ＪＯＢ名         */
            DCL  VAR(&PRMUSR) TYPE(*CHAR) LEN(10)  /*  ユーザー名       */
            DCL  VAR(&PRMJNO) TYPE(*CHAR) LEN(6)   /*  番号             */
            DCL  VAR(&PRMSFN) TYPE(*DEC)  LEN(6 0) /*  ＳＰＬＦ番号     */
            DCL  VAR(&PRMJBN) TYPE(*CHAR) LEN(8)   /*  ＪＯＢシステム名 */

            /* 保留 */
                 IF         COND(&PRMKBN *EQ '3')  +
                    THEN(DO)
                     HLDSPLF FILE(&PRMSPF) +
                     JOB(&PRMJNO/&PRMUSR/&PRMJOB) SPLNBR(&PRMSFN)
                     MONMSG MSGID(CPF0000)
                    ENDDO

            /* 削除 */
                 IF         COND(&PRMKBN *EQ '4')  +
                    THEN(DO)
                     DLTSPLF FILE(&PRMSPF) +
                     JOB(&PRMJNO/&PRMUSR/&PRMJOB) SPLNBR(&PRMSFN)
                     MONMSG MSGID(CPF0000)
                    ENDDO

            /* 表示 */
                 IF         COND(&PRMKBN *EQ '5')  +
                    THEN(DO)
                   CPYSPLF    FILE(&PRMSPF) TOFILE(*LIBL/WFR10HPF) +
                                JOB(&PRMJNO/&PRMUSR/&PRMJOB) +
                                SPLNBR(&PRMSFN) TOMBR(&PRMMBR)
                     MONMSG MSGID(CPF0000)
                    ENDDO

            /* 送信 */
                 IF         COND(&PRMKBN *EQ '6')  +
                    THEN(DO)
                     RLSSPLF FILE(&PRMSPF) +
                     JOB(&PRMJNO/&PRMUSR/&PRMJOB) SPLNBR(&PRMSFN)
                     MONMSG MSGID(CPF0000)
                    ENDDO

ENDPGM
```